
Introduction
The Single Line Edit (SLE) is one of the most basic controls at a programmer’s disposal.  The purpose of this
paper is to show how the basic SLE can be expanded.  The following topics will be covered:

 Subclassing

 Basic Dispatching

 Effectively Using Buttons with SLEs

 Optimizing

 Specialized Subclasses

The source code to this session will be available on the CA web site, as well as my web site
(http://www.wmconsulting.com).

Subclassing

Why Subclass?

Subclassing performs a couple of functions. First, it protects your code. The next version of CA-Visual Objects
may change some functionality of the SLE. By subclassing, I can ensure that my default behavior will always
happen. Secondly, subclassing allows you to expand a class without changing the base class.

Creating

The first step in extending the SingleLineEdit class functionality is to create a subclass. To create the subclass,
add this line of code to a module in your project:

CLASS mySLE INHERIT SingleLineEdit

Abstract Class

Once you have created your subclass, the base class should be considered an abstract class, that is to say, a class
that is not designed to be called directly. Always use your subclass so keystrokes, accesses, etc. can be handled
uniformly across your application.  The following is an example of how the SingleLineEdit class can be broken
down.
SingleLineEdit >>>>>>Abstract Class (do not call directly)

RightSLE >>>>>>Container class (major keyhandling goes here)
FileSLE |

pbFileSLE |
DateSLE | >>>>>>Subclasses that are specialized
SearchSLE |
CalcSLE |

pbCalcSLE |

By having a single subclass of SingleLineEdit that acts as a base class for specialized subclasses like searchSLE
and calcSLE, Keys like the Esc key and Enter key will act the same way across the subclasses.

FieldSpecs

Fieldspecs are very useful, as you can define a few that will handle most of your data types. The biggest benefit
of using fieldspecs is that you can change your database design with minimal code changes. For example, if your
application uses a lot of numeric 10.2 fields and you get a customer request to increase the size from 10.2 to
12.2, this becomes a simple change. By changing your numeric fieldspec to 12.2, all of your datawindows will
be changed without recoding.

You can subclass the SingleLineEdit control without using fieldspecs.  But the fieldspec makes it easier to code,
as there are properties that you can test for and against.

Using

To use your subclass, or any of its specialized subclasses, simply use the "inherit from" property and select the
subclass that you want to place (i.e., rightSLE, searchSLE, etc). Please note that you have to compile your
subclass BEFORE it will appear in the drop down list for the “inherit from” property box.

Basic Dispatching
You can attach a dispatcher to any control. This gives the programmer incredible control through the ability to
catch and handle events before normal processing occurs. The following is a dispatcher at its most basic level:
Class mySLE inherit SingleLineEdit

Method dispatch() class mySLE
Infoxbox{self,”Single Line Edit”,”I’m in the dispatcher”}:show()
Return super:dispatch()

In this sample, a message is displayed, and then normal processing occurs (return super:dispatch() takes care of
this). However, there are times where you do not want normal processing to occur.

Stopping Further Processing

The way to stop further processing from inside a dispatcher is to return 1L. This tells Windows that the event has
been taken care of and no further action is required.

When Is Enough Enough?

Computer Associates sells a companion product to CA-Visual Objects called the CA-Visual Objects SDK
(Software Developers Kit). If you attended Technicon last year, you already have a copy of it. This is an
invaluable tool. It contains the source to the GUI classes. I recommend that you use the SDK to see what the
developers’ intentions were with a given class. That way you can make intelligent choices on when it is
appropriate to leave a dispatcher without calling the super.

Responding to the Enter Key

One of the first things I wanted to do when I started programming in Windows was to make the keyboard
consistent with CA-Clipper’s. This meant that I had to have the Enter key take me between controls. CA-Visual
Objects does provide this functionality via the CLIPPERKEYS property of the Window Editor. However, using
CLIPPERKEYS does have some unwanted side effects, the biggest being that pressing Enter over a push button
takes you to the next field. But, by using inheritance, we can subclass the Single Line Edit and overcome that
limitation. The key to example 2 is the KEYUP method. That is where you trap the Enter key.

CLASS mySLE INHERIT SingleLineEdit

PROTECT INSTANCE hOwner AS WORD
Protect nDirection AS INT
Protect lEnterKeyMoves AS LOGIC // gets assigned from

// CAVOWED.INF

METHOD Init( oOwner, nId ) CLASS mySLE

Super:Init(oOwner, nID)

// Get Windows handle of owner window

IF IsInstanceOf( SELF:Owner, #DataWindow )
SELF:hOwner := SELF:Surface:Handle()

ELSE
SELF:hOwner := SELF:Handle()

ENDIF
SELF:nDirection := 1 // go forward

METHOD KeyUp(oE) CLASS mySLE

// Process keys as required

DO CASE

CASE oE:KeyCode == KEYENTER
If SELF:lEnterKeyMoves

If nDirection
Do case

CASE nDirection = 1

// Set focus to next control on window

PostMessage( SELF:hOwner, WM_NEXTDLGCTL, 0, 0L )
CASE nDirection = 2

// move to previous control on window

postMessage( SELF:hOwner, WM_NEXTDLGCTL, 1, 0L )
END CASE

ENDIF
ENDIF

ENDCASE

RETURN SUPER:KeyUp(oE)

Assign EnterKeyMoves(uNew) class mySLE
Self:lEnterKeyMoves := uNew
Return self:lEnterKeyMoves

Window Editor Modifications

With the Enter key taken care of, it would be nice to have a property on the Window Editor designer so you can
turn the Enter key on and off. Fortunately, that is an easy fix. The file CAVOWED.INF has to have a couple of
lines added to it and bingo, new property.

[CONTROL:TEXTCONTROL:EDIT:SINGLELINEEDIT]
Assign01=Block,Block(string)
Assign02=Block Owner,BlockOwner(STRING)
Assign03=OverWrite,OverWrite(OVERWRITE)
Assign04=Picture,Picture(STRING)
Assign05=Scroll Mode,ScrollMode(SCRMODE)
Assign06=Focus Select,FocusSelect(FSELECT)
Assign07=Auto Focus Change,AutoFocusChange(BOOL)
Assign08=Move on EnterKey,EnterKeyMoves(BOOL)
ContTitle=Single-Line Edit
ClassName=edit
Name=SingleLineEdit
Size=80,20
Style=WS_CHILD|WS_TABSTOP|WS_BORDER|ES_AUTOHSCROLL
ExStyle=WS_EX_CLIENTEDGE
HelpID=2185
PropTabs=HyperLabel,rightSLE:General,SingleLineStyles:Styles,ExStyles
InitMethod=SingleLineEdit(PARENT,ID,POSITION,SIZE)
InheritClassName=Inherit from Class,InheritClassName(CLASS:SingleLineEdit)

In the Assignmap section, add the following to the end of the section:

EnterKeyMoves=

And finally, in the stdProperties section, add the following to the end of the section:

rightSLE=(General SinglelineEdit Properties)Move on EnterKey,Block,Block
Owner,Field Spec,Text Limit,Text Color,Background Color,Inherit from
Class,Context Menu,Tooltip,Font,Picture,Overwrite,Scroll Mode,Focus
Select,Auto Focus Change,Input Method Editor,_Visible,_GenCode,Owner
Alignment,_Width,_Height,_Left,_Top

NOTE: Always make changes to CAVOWED.INF with great caution. While the above sample will indeed let
you add an additional property to the single Line Edit, it can cause unwanted side effects. For example, after
making this modification, any Single Line Edit control that is added that does not inherit from your subclass will
cause a runtime error, as it does not have the access/assign to handle the Enter key. But even this can be
overcome. You can always add your own access/assign to the Single Line Edit class itself. Unfortunately, if you
do this, you will still not have the code to handle the key, just to suppress the error message.

Effectively Using Buttons with SLEs
Buttons are a visual aid that many end-users “demand”. Unfortunately, CA-Visual Objects does not natively
support a buttonSLE class. However, subclassing can again come to our aid. By starting off with a button
abstract class, we can subclass to create buttons that respond to each of our SLEs. There are a couple of design
considerations that must be addressed.

 Must be easy to implement

 Must respond to messages just as the SLE does

 Must integrate into the IDE

Easy to Implement

The button must be easy to implement. This is accomplished through the “inherit from” property of the IDE.
Each specialized subclass should be named in a way that makes it easy to identify. For example, the fileSLE
pops up a file open dialog box when double clicked. The pushbutton version is called pbfileSLE.

Must Respond to Messages

The pushbutton classes must respond to messages such as hide/show. They must also obey the readonly status.

Must Integrate into the IDE

A pushbutton subclass must integrate well into the IDE. In other words,  the visual layout of the SLE should not
change when we add the pushbutton. The code below demonstrates that principle. The size of the SLE is shrunk
down to accommodate the pushbutton. In that way, the layout of surrounding controls is not affected.

METHOD Init(oForm,oResID,oPoint,oDim,kStyle) CLASS pbFileSLE
LOCAL lpRect IS _WINRECT
SUPER:init(oForm,oResID,oPoint,oDim,kStyle)
oPoint := SELF:origin
oDim := SELF:size
oPoint:y -= 1
oPoint:x := oPoint:x + oDim:width-28
oDim:width := 30
oDim:height += 1
oPB := filePB{oForm,10102,oPoint,oDim,SELF}
SetWindowPos(SELF:handle(),0L,0,0,SELF:size:width-30,SELF:size:height,SWP_NOMOVE+0x14)
GetWindowRect(SELF:handle(),@lpRect)
// before we leave, show the push button we created
oPB:show()

RETURN SELF

Optimizing Your SLE
There are many different ways in which you can optimize your SLE controls. This next section will take a look
at the various specialized subclasses that comprise rightSLE and show how we can make the code a bit more
maintainable.

Buttons

In the last section, we looked at buttons and how we could use them to make an SLE more intuitive.
Unfortunately, the code that we end up with involved many cuts and pastes. While at first this appears to be the
quickest way to get code up (it really is ), it increases the backend cost of maintaining your code. With that in
mind, let’s rethink the button class.

Rethinking

Each specialized button class involves four basic elements:

1. Button up

2. Button down

3. Message to send

4. SLE that it interfaces with

With this in mind, we can abstract back one layer and build this into an SLEButton class that handles the dirty
work for us. The following code demonstrates this new button class.

CLASS SLEButton INHERIT wmBitmapButton
PROTECT oJoinedSLE AS OBJECT
PROTECT nMessage AS DWORD

METHOD Init(oForm,oResID,oPoint,oDim,oSLE) CLASS SLEButton
SELF:oJoinedSLE := oSLE
SUPER:init(oForm,oResID,oPoint,oDim,,WS_TABSTOP)
nMessage := WM_LBUTTONDBLCLK // default message to send

METHOD Dispatch(oE) CLASS SLEButton
DO CASE

CASE oE:Message == WM_LBUTTONUP
// I use postmessage here so I do not wait for
// the dialog to close before I finish handling
// this button press

PostMessage(SELF:oJoinedSLE:handle(),SELF:nMessage,0,0L)
RETURN 1L

ENDCASE
RETURN( SUPER:Dispatch(oE) )

The Main Subclass

The next section to look at is the main subclass of  Single Line Edit. This is really a container class. Its job is to
protect the behavior that we implement. It processes all SLE keystrokes and passes off to the super anything that
we do not explicitly handle. With that in mind, the first opportunity for optimization is the statement:

If !self:Readonly
// some code goes here

endif

By moving this to the front of the main case structure, we can improve performance and readability. By
improved performance, I am not speaking of huge time savings. However, if you have clients on older machines
(486/ Pentium 75,100,133,166), they will appreciate anything you can give them. Readability is improved by one
less if/endif to look through when debugging or doing code reviews. Just remember that if you have a dispatch in
any of your specialized classes, you should check for readonly at the beginning of its dispatch. Here is an
example of the modified case:

METHOD Dispatch ( oEvent ) CLASS rightSle

/*
Only the messages, that rightSle wants to know about.
All the rest go through to SingleLineEdit's dispatch and every other
class's dispatch that is involved from there - There are lots of
them!!!!

*/
DO CASE

CASE SELF:ReadOnly
// we don't want to do anything

CASE !(SELF:FieldSpec == NULL_OBJECT ) .and. ;
SELF:FieldSpec:Valtype == 'D'
DO CASE

// Support for VK_ADD and VK_SUBTRACT like in Quicken
CASE oEvent:Message == WM_KEYDOWN .And. ;
( oEvent:wParam == VK_ADD .Or.;
oEvent:wParam == VK_SUBTRACT )
// Grab the event, the KeyDown and send it over to
// PEDateSle's KeyDown

SELF:CalKeyDown( ControlEvent{ oEvent } )
RETURN 1L

// Stop the f&^*%$ thing beeping
CASE oEvent:Message == WM_CHAR .And. ;

( CHR( oEvent:wParam ) == "+" .Or. ;
CHR( oEvent:wParam ) == "-" )

// Beep if you have too, but don't process
// either of these keys

IF CToD( SELF:TextValue ) == NULL_DATE
MessageBeep( 0xFFFFFFFF )

ENDIF

/*
By returning 1L we prevent
oEditString:ProcessKeyEvent() being
called in the dispatch of
SingleLineEdit..
ProcessKeyEvent() of __FormattedString
calls ProcessChar() which in turn calls
MatchesTemplChar() which will fail
because of the + or - . When it fails it
calls InvalidAction() of _FormattedString
which is the MesssageBeep which was
Driving me up the wall. Comment out the
above lines to see ( hear ) what I was
complaining about.

*/

RETURN 1L
CASE oEvent:Message == WM_COMMAND
IF oEvent:wParam == ;

IDM_PEDateSleContextMenu_File_Calendar_ID
// Pop up the calendar
SELF:ShowCalendar( ControlEvent{ oEvent } )
RETURN 1L

ELSEIF oEvent:wParam == ;
IDM_PEDateSleContextMenu_File_Today_ID

// Just stuff today's date back into
// PEDateSle
SELF:Value := Today()
RETURN 1L

ENDIF
END CASE

CASE !(SELF:FieldSpec == NULL_OBJECT ) .and. ;
SELF:FieldSpec:Valtype == 'N'

DO CASE
CASE oEvent:Message == WM_LBUTTONDBLCLK

SELF:SetFocus()
SELF:selection := selection{0,-1}
SELF:lAllSelected := TRUE
RETURN 1L

CASE oEvent:Message == WM_KEYDOWN
DO CASE
CASE oEvent:wparam == VK_BACK .or. ;

oEvent:wparam == VK_END .or. ;
oEvent:wparam == VK_HOME .or. ;
oEvent:wparam == VK_LEFT .or. ;
oEvent:wparam == VK_RIGHT .or. ;
oEvent:wParam == VK_DELETE
SELF:processNum( ControlEvent{ oEvent } )
IF oEvent:wparam == VK_BACK .or.;

oEvent:wParam == VK_DELETE //
movement keys other than
backspace need to finish
processing
RETURN 1L

ENDIF
ENDCASE

CASE oEvent:Message == WM_CHAR
// Grab the event, the KeyDown and send it over
// to rightSle's KeyDown

SELF:processNum( ControlEvent{ oEvent } )
RETURN 1L

ENDCASE
END CASE

// Let all the other messages, go back to VO's Sle
Wndproc

RETURN( SUPER:Dispatch( oEvent ) )

Duplication of Code

Most programmers are creatures of habit. Cut and paste is quick, and easy. However, making of habit of cut and
paste can lead to maintenance problems. Think about this situation. You discover a “bug” in some of your code.
Unfortunately, you have cut and pasted this code (3 or 4 lines) into 30 or 40 places across several methods. It is
not going to be a fun afternoon finding and correcting each error.

If you find yourself cutting and pasting a lot: Stop. Think  about why you are cutting and pasting. The lines that
you are cutting and pasting are probably a very good candidate for a new method. That way we do not end up
with duplication of code. A side benefit is that the code is easier to read and maintain.

Early Binding

CA-Visual Objects supports “early binding”. Early bound means that the compiler can resolve everything about
the instance variable at compile time. Early bound instance variables are generally faster as they do not have to
be resolved at runtime. CA-Visual Objects also supports the early binding of methods. Because early bound
instance variables and methods are resolved at compile time, the compiler will enforce data-type rules and return
values. This will result in fewer runtime errors! To take advantage of early binding, we must declare our
methods and instance variables to the compiler. Here is what our SLE class might look like:

CLASS rightSle INHERIT SingleLineEdit
PROTECT oPB AS OBJECT
PROTECT lTurnNegative AS LOGIC
PROTECT lAllSelected AS LOGIC
EXPORT lAlignCalendarLeft AS LOGIC // Used to make

// calendar align either
// left or right on the sle

EXPORT FirstDayoftheWeek AS DWORD // for the dataSLE

DECLARE ACCESS zeropad
DECLARE METHOD processnum, setfield
DECLARE METHOD ShowCalc, ShowCalendar

Speciality Subclasses
With our “container” class optimized, we can now extend it to create specialty subclasses. The main class
contains the logic for processing right aligned numbers and calendar controls.  The following classes will be
mentioned:

 SearchSLE

 File Dialog Class

 DateSLE

 Calculator Class

SearchSLE
The SearchSLE is a subclass of SingleLineEdit that allow you to set up a browser window and pass the fields
and dataserver at runtime. Below is the init of SearchSLE.

METHOD Init(oForm,oResID,oPoint,oDim,kStyle) CLASS SearchSLE
LOCAL lpRect IS _WINRECT
SUPER:init(oForm,oResID,oPoint,oDim,kStyle)
SELF:symMethod := String2Symbol("sleSearch")
oPoint := SELF:origin
oDim := SELF:size
oPoint:y -= 1
oPoint:x := oPoint:x + oDim:width-28
oDim:width := 30
oDim:height += 1
SELF:oPB := SLEButton{oForm,-1,oPoint,oDim,SELF}
SELF:oPB:hBitmapUp := LoadBitmap(_GetInst(),String2Psz("IBM_MAGNIFUP"))
SELF:oPB:hBitmapDown := LoadBitmap(_GetInst(),String2Psz("IBM_MAGNIFDN"))
SetWindowPos(SELF:handle(),0L,0,0,SELF:size:width-30,SELF:size:height,SWP_NOMOVE+0x14)
GetWindowRect(SELF:handle(),@lpRect)
// before we leave, show the push button we created
SELF:oPB:show()

The key to SearchSLE is the following line:
SELF:symMethod := String2Symbol("sleSearch")

This sets up an exported variable called symMethod, which is the method to invoke when the pushbutton is
clicked. A double-click on the SLE will also invoke the symMethod.

The Dispatch

The trick to sending the SLE to the method is contained in the SearchSLE’s dispatch. It uses the send command
to invoke the method (if it exists). Here is the code to invoke the method.

METHOD Dispatch(oEvent) CLASS SearchSLE

DO CASE

CASE SELF:ReadOnly

// we don’t want to search a readonly field

CASE oEvent:Message == WM_LBUTTONDBLCLK

IF IsMethod(SELF:owner,SELF:symMethod)

Send(SELF:owner,SELF:symMethod,SELF,SELF:CurrentText)

ENDIF

RETURN 1L

ENDCASE

RETURN( SUPER:Dispatch(oEvent) )

You will notice that the IsMethod function is used to check for the existence of the method before it is called.
This is necessary, as the symMethod variable is not known at compile time.

How Do I Use It?

SearchSLE is really quite simple to implement. It is easiest to invoke it during the postinit method of your data
window. The following code shows how to hook a searchSLE.

METHOD PostInit(oWindow,iCtlID,oServer,uExtra) CLASS dwTest
//Put your PostInit additions here
self:oDCsleSearchme:symMethod := String2Symbol("mySearch")
RETURN NIL

METHOD mySearch(oSLE,cSLECurrentTextValue) CLASS dwTest
Infobox{self:owner,”Info”,”In a custom search method”}:show()

RETURN NIL

FileSLE
Microsoft provides a set of standard dialogs for doing common tasks. One of those tasks is a standard dialog for
selecting a file. By subclassing SingleLineEdit, we can take advantage of the standard File Open dialog and send
the results back to our Single Line Edit. The following example shows how we can accomplish this task.

CLASS FileSLE INHERIT rightSLE
PROTECT nDialogType := 1 AS INT
EXPORT xFilter,xFilterDesc AS USUAL
DECLARE METHOD ShowFileDialog

METHOD Dispatch(oEvent) CLASS FileSLE
DO CASE

CASE oEvent:Message == WM_LBUTTONDBLCLK
SELF:showFileDialog()
RETURN 1L

ENDCASE
RETURN( SUPER:Dispatch(oEvent) )

METHOD showFileDialog() AS VOID PASCAL CLASS fileSle
LOCAL oFileDLG AS StandardFileDialog
IF nDialogType == 1 // Open File Dialog

oFileDLG := OpenDialog{SELF:owner,SELF:currenttext}
IF SELF:xFilter <> NIL

oFileDLG:SetFilter(SELF:xFilter,SELF:xFilterDesc)
ELSE

oFileDLG:SetFilter("*.*","All FIles")
ENDIF
oFileDLG:show()
IF !Empty(oFileDLG:FileName)

SELF:textvalue := oFileDLG:FileName
ENDIF
SetFocus(SELF:handle())

ELSE
oFileDLG := SaveAsDialog{SELF:owner,SELF:currenttext}
IF SELF:xFilter <> NIL

oFileDLG:SetFilter(SELF:xFilter,SELF:xFilterDesc)
ELSE

oFileDLG:SetFilter("*.*","All FIles")
ENDIF
oFileDLG:show()
IF !Empty(oFileDLG:FileName)

SELF:textvalue := oFileDLG:FileName
ENDIF
SetFocus(SELF:handle())

ENDIF
RETURN

pbFileSLE

In addition to the fileSLE, there is pbfileSLE. This is a pushbutton version of the File Open dialog that inherits
from fileSLE. It places a Bitmap Button of a file folder immediately beside the SLE. This give a visual clue as to
what input is expected. It also keeps you from having to double-click the SLE to open up the dialog box.
However, the pbfileSLE does respond to the double click just as its parent does.

DateSLE
DateSLE is a subclass of PEDateSLE written by Graham McKenie. It adds a Bitmap Button of a calendar
immediately to the right of the SLE. It does maintain the right mouse menu of PEDateSLE. The following is the
init of dateSLE.

METHOD Init(oOwner, nId, oPoint, oDim, kStyle ) CLASS DateSle
LOCAL lpRect IS _WINRECT
LOCAL DIM firstDay[2] AS BYTE

SUPER:Init( oOwner, nID, oPoint, oDim, kStyle )
oPoint := SELF:origin
oDim := SELF:size
oPoint:y -= 1
oPoint:x := oPoint:x + oDim:width-28
oDim:width := 30
oDim:height += 1
SELF:oPB := SLEButton{oOwner,-1,oPoint,oDim,SELF}
SELF:oPB:hBitmapUp := LoadBitmap(_GetInst(),String2Psz("IBM_CALUP"))
SELF:oPB:hBitmapDown := LoadBitmap(_GetInst(),String2Psz("IBM_CALDN"))
SELF:oPB:show()
SetWindowPos(SELF:handle(),0L,0,0,SELF:size:width-30,SELF:size:height,SWP_NOMOVE+0x14)
GetWindowRect(SELF:handle(),@lpRect)

SELF:FieldSpec := DateFS{}

// Add a popup menu to the PEDateSle - two items Today and Calendar
SELF:ContextMenu := PEDateSleContextMenu{ SELF )

// Default to an empty data value, if you don't the keydown method will screw up
// because the test of If ! (oControl:Value == NULL_DATE ) will fail because
// oControl:Value will be Nil not Null_Date
SELF:Value := Null_Date
GetLocaleInfo(LOCALE_SYSTEM_DEFAULT, LOCALE_IFIRSTDAYOFWEEK, @firstDay, 2)
IF firstDay[1] = 48 // Monday 1st Day of Week

SELF:FirstDayoftheWeek := 2 // set up Sunday as initial day of the week
ELSEIF firstDay[1] = 54 // Sunday 1st Day of Week

SELF:FirstDayoftheWeek := 1 // set up Sunday as initial day of the week
ENDIF

The last few lines of the init are probably the most interesting; they set up the first day of the calendar. Not all
countries use Sunday as the first day of the week. Asking windows for the first day of the week means that the
calendar will perform correctly wherever you install it.

Sending a Date to the Calendar

One handy addition to the calendar class is the ability to pass whatever is in the SLE to the control and set the
calendar . The default is to always set the calendar to today(). To accomplish this, the following postinit() was
used instead of the original  postinit() of dlgPECalendar.

METHOD PostInit(oParent,uExtra) CLASS dlgPECalendar
LOCAL oOrigin AS Point
LOCAL oSize AS Dimension

LOCAL lAlignLeft AS LOGIC
LOCAL oDateSle AS OBJECT
LOCAL dOldDate AS DATE

oDateSle := uExtra
// pick up what the user typed into the SLE control

dOldDate := CToD(oDateSle:textvalue)

// Get the stuff about the sle that we need
oOrigin := PClone( oDateSle:Origin )
oSize := PClone( oDateSle:Size )
lAlignLeft := oDateSle:lAlignCalendarLeft

// Origin:x is always the Sle's origin:x unless the alignment is right-

oOrigin:x := iif( ! lAlignLeft, oOrigin:x + oSize:Width, oOrigin:x )

// Origin:y has to be fiddled with to make the calendar align just below the
Sle. Jeez I hate Cartesian coorordinates

// - one line of code and how long does it take to figure it out. Bloody
CommonView, why didn't they take it all out

SELF:Origin := Point{ oOrigin:x, SELF:Owner:Size:Height - SELF:Size:Height -
( SELF:Owner:Size:Height - oOrigin:y ) }

SELF:oDCCalendar:CurrentDate := iif(dOldDate == NULL_DATE,Today(),dOldDate)
SELF:oDCCalendar:FirstDayoftheWeek := oDateSLE:FirstDayoftheWeek

RETURN( SELF )

Conclusion
This paper has presented several extensions to the default SingleLineEdit. CA-Visual Objects is a very powerful
language, and through inheritance, you can make your SingleLineEdits much more user friendly.

